> restart:
Specifiy the number of control volumes. The smallest number is N = 3
> Ncv:=5:
> N := Ncv:
Specify the geometric and thermophysical parameters.
> fin1:=[a=4.6/1000,b=4.6/1000,L=100/1000,k=100,h=40,he=10^(-20),hc=10^(20),Tb=100,Tf=25,N=Ncv]:#Circular profile
Specify the base and fluid temperatures.
> Temps:=[Tb=100,Tf=25]:
The local conduction area.
> A:=Pi*a*a:
> Aj:=evalf(subs(fin1, A));

Dimensionless convection surface area for each control volume.
> Sjstar:=(2*Pi*a*L)/Ncv:
Convection surface area for each control volume.
> Sj:=evalf(subs(fin1, Sjstar));

Conduction at fin base.
> Qbase:=hc*A*(Tb-Tf-theta(0)):
Convection at fin tip.
> Qend:=he*A*theta(N+1):
Conduction from base into the base control volume.
> Q0:=k*A*(theta(0)-theta(1))/(L/(2*N)):
Conduction out of tip control volume through tip.
> QN:=k*A*(theta(N)-theta(N+1))/(L/(2*N)):
Conduction at internal control volume surfaces.
> Qj:=j->k*A*(theta(j)-theta(j+1))/(L/N):
Convection from control volumes.
> Qfj:=j->h*Sj*theta(j):
Create a list of excess temperatures nodes for  j = 0 through  j = N + 1.The are N + 2. 
> thetas:={seq(theta(j),j=0..Ncv+1)}:
Heat balance at the base surface.
> HBbase:=Qbase-Q0=0:
Heat balance for the first control volume.
> HB1:=Q0-Qj(1)-Qfj(1)=0:
Heat balance relationship for all internal control volumes.
> HBj:=j->Qj(j-1)-Qj(j)-Qfj(j)=0:
Heat balance for the last control volume.
> HBN:=Qj(N-1)-QN-Qfj(N)=0:
Heat balance at the fin tip.
> HBend:=QN-Qend=0:
Create the list of equations for the excess temperatures from the heat balance relationships.
> eqs:=subs(fin1,Temps,{HBbase,HB1,seq(HBj(j),j=2..Ncv-1),HBN,HBend}):
Solve the equations for the excess temperature nodes from theta[0] to theta[N + 1].
> temps1:=solve(eqs,thetas);

> temps2:=evalf(%):
Assign the excess temperature nodes.
> assign(temps2):
Create a list of the temperatures from the relation T[j] = T[f] + theta[j].
> Tjs:=evalf(subs(Temps,[seq(T(j)=theta(j)+Tf,j=0..Ncv+1)]),4);

> Te:=evalf(subs(Temps,seq(theta(j)+Tf,j=Ncv+1)),4):
Calculate the fin heat transfer rate by conduction at the base surface.
> Qfin:=k*Aj*(theta(0)-theta(1))/(L/(2*N)):
> Qfin1:=evalf(subs(N=Ncv,fin1,Qfin));

Calculate the fin heat transfer rate through the base interface. This value will be equal to the above value.

Calculate the fin resistance.
> Rfin1:=evalf(subs(Temps,(Tb-Tf)/Qfin1)):
Calculate the heat dissipation by convection  from the sides of the spine.
> Qsides1:=evalf(subs(N=Ncv,fin1,add(h*Sj*theta(j),j=1..Ncv)));

Calculate the heat transfer by convection through the end surface.
> Qend1:=evalf(subs(fin1,he*Pi*a^2*theta(Ncv+1))):
Calculate the total heat dissipation by convection from all surfaces. 
> Qfin2:=Qsides1+Qend1:
Calculate the ideal (maximum) heat dissipation from the fin. Assume perfect base contact.
> Qideal:=(h*2*Pi*b*L+he*Pi*a^2)*(Tb-Tf):
> Qideal1:=evalf(subs(Temps,fin1,Qideal));

Calculate the fin efficiency for perfect base contact.
> eta1:=evalf(Qfin1/Qideal1,6);

Summary of calculations.
> summary:=evalf([Ncv,Qi=Qideal1,Qf=Qfin1,Rf=Rfin1,effy=eta1],6);fin1;

> Qs:=evalf(subs(fin1, {Qbase, Qend, QN, Q0, seq(Qj(j), j = 1 .. Ncv-1), seq(Qfj(j), j = 1 .. Ncv-1)})):
Entropy Balance at the base
> Sgen0:=-(Qbase/(Tb+273))+(Q0/(T(0)+273)):
Entropy Balance at nose 1
> Sgen1:=-(Q0/(T(0)+273))+(Qj(1)/(T(1)+273))+(Qfj(1)/(T(1)+273)):
General Enropy Balance where j=1 to Ncv-1
> Sgenj:=j->-(Qj(j-1)/(T(j-1)+273))+(Qj(j)/(T(j)+273))+(Qfj(j)/(T(j)+273)):
Entropy balance at node N
> SgenN:=-(Qj(Ncv-1)/(T(Ncv-1)+273))+(QN/(T(Ncv)+273))+(Qfj(Ncv)/(T(Ncv)+273)):
Entropy Balance at the end
> Sgenend:=-(Qj(Ncv)/(T(Ncv)+273))+(Qend/(T(Ncv+1)+273)):
Sgens
> Sgens:={seq(Sgen(j),j=0..Ncv)}:
Equation balance equations
> EQNSSB:=evalf(subs(fin1,Tjs,{seq(Sgenj(j),j=2.. Ncv-1),Sgen0,Sgen1,SgenN,Sgenend}));

Entropy Generation Evaluation
> TSgen:=evalf(subs(fin1,Tjs,[Sgen0+Sgen1+SgenN+Sgenend]+[sum(Sgenj(j),j=2.. Ncv-1)]));

